

This presentation and associated control stream examples are available from
http://www.page-meeting.org/default.asp?id=26&keuze=abstract-view&goto=abstracts&orderby=author&abstract_id=838

Slide
1

PKPD

Workshop

Nick Holford
Dept Pharmacology & Clinical Pharmacology
University of Auckland, New Zealand

http://www.page-meeting.org/default.asp?id=26&keuze=abstract-view&goto=abstracts&orderby=author&abstract_id=838

Slide
2



PK=Pharmacokinetics
PD=Pharmacodynamics
Rx=Prescription (Latin *recipe*)

Slide
3


Warfarin Data

- PKPD Studies in Healthy Subjects
 - 1.5 mg/kg single oral dose
 - Total racemic warfarin plasma concentration
 - Prothrombin complex activity (PCA)
- 32 subjects, 250 concentrations, 232 PCA
- O'Reilly RA, Aggeler PM, Leong LS. Studies of the coumarin anticoagulant drugs: The pharmacodynamics of warfarin in man. *Journal of Clinical Investigation* 1963;42(10):1542-1551
- O'Reilly RA, Aggeler PM. Studies on coumarin anticoagulant drugs: Initiation of warfarin therapy without a loading dose. *Circulation* 1968;38:169-177

A classical clinical pharmacology study from 40 years ago. Prothrombin complex activity is inversely proportional to the International Normalized Ratio (INR)

Slide
4

Warfarin Observations




©NHQ Hafford, 2005, all rights reserved.

Two features to notice about the time course of warfarin concentration and effect on PCA:
Peak concentration is about 6 h while greatest effect is at 72 h
Most marked between subject variability in PCA occurs when PK variability is least

Slide
5

Warfarin First-Order Input (KA1L)




©NHQ Hafford, 2005, all rights reserved.

Individual "post hoc" predictions of concentration fit the time course well and should be adequate for driving the pharmacodynamic model.

Slide
6

Warfarin KA1L Predictive Check



©NHQ Hafford, 2005, all rights reserved.

The visual predictive check reveals that the model overestimates the early variability in concentration.

Slide
7

Immediate Effect Model

- Two Basic Approaches
 - \$PRED
 - Write model for CP
 - PREDPP
 - ADVAN2
 - ADVAN6 Differential Equations

©1990 Habbert, 2005, all rights reserved.

An immediate effect model uses central compartment concentrations to predict the drug effect. Concentrations can be predicted using \$PRED or one of the PREDPP library ADVAN subroutines. Using ADVAN subroutines is usually easier to code but the model is not so clear.

Slide
8

Central Compartment Using ADVAN2

```
SSUB ADVAN2 TRANS2      ; Define parameters
SPK                      KA=LN2/TABS
; Define LN2              ALAG1=TLAG
IF (NEWIND.LE.1) THEN     S2=V
    LN2=LOG(2)            V2=V
ENDIF                    $ERROR

CP=A(2)/V ; or CP=F
IF (DVID.LE.1) THEN
    Y=CP*(1+RUVCV) + RUVSD
ENDIF

CE=CP
PCA=E0 + EMAX*CE/(C50+CE)
IF (DVID.EQ.2) THEN
    Y=PCA + RUVFX
ENDIF
```

©1990 Habbert, 2005, all rights reserved.

TABS is the half-life of absorption. The absorption rate is parameterised in terms of TABS but because ADVAN2 requires a value for KA it must be calculated from TABS in the code. The NEWIND variable is a built in feature of NONMEM. When it is <= 1 it means this is the first record for this individual. The variable LN2 is calculated from LOG(2) just once for efficiency. Note how the DVID data item is used to distinguish predictions of concentration (DVID=1) from predictions of PCA (DVID=2).

Slide
9

Data for Joint PKPD Model

#ID	time	wt	age	sex	amt	dvid	dv	mdv
0	0	66.7	50	1	100	0		1
0	0	66.7	50	1		2		1
0	0.5	66.7	50	1		1	0	0
0	1	66.7	50	1		1	1.9	0
0	2	66.7	50	1		1	3.3	0
0	3	66.7	50	1		1	6.6	0
0	6	66.7	50	1		1	9.1	0
0	9	66.7	50	1		1	10.8	0
0	12	66.7	50	1		1	8.6	0
0	24	66.7	50	1		1	5.6	0
0	24	66.7	50	1		2	44	0
0	36	66.7	50	1		1	4	0
0	36	66.7	50	1		2	27	0
0	48	66.7	50	1		1	2.7	0
0	48	66.7	50	1		2	28	0

©1990 Habbert, 2005, all rights reserved.

NONMEM requires that all observations are in a single DV column. In order to distinguish concentrations from effects (PCA) the DVID data item is used to identify the kind of DV. A DVID of 0 is used to indicate a dose record, 1 for a concentration record and 2 for a PCA record. Note that MDV is used to indicate if there are missing observations e.g. this subject does not have a PCA observation at TIME=0.

Slide
10

Central Compartment Using \$PRED

```
$PRED
; Get DOSE from AMT
IF (NEWIND.LE.1) THEN
  LN2=LOG(2)
  DOSE=0
ENDIF
IF (AMT.GT.0) DOSE=AMT

; Define parameters
KA=LN2/TABS
KE=CL/V

; Adjust time for lagtime
IF (TIME.LE.TLAG) THEN
  TNOW=0
ELSE
  TNOW=TIME-TLAG
ENDIF

; Plasma concentration
EXPKA=EXP(-KA*TNOW)
EXPKE=EXP(-KE*TNOW)
CP=DOSE*KA/(V*(KA-KE))*(EXPKE-EXPKA)

IF (DVID.LE.1) THEN
  Y=CP*(1+RUVCV) + RUVSD
ENDIF

CE=CP
PCA=E0 + EMAX*CE/(C50+CE)
IF (DVID.EQ.2) THEN
  Y=PCA + RUVFX
ENDIF
```

©NHQ Holford, 2005, all rights reserved.

Note that DOSE must be defined on every record so it is obtained 'on the fly' from the AMT value. The AMT value is only >0 at TIME=0.

Slide
11

Central Compartment Using ADVAN6

```
$SUBR ADVAN6 TOL=3
$MODEL
COMP (DEPOT)
COMP (CENTRAL)
$PK
; Define LN2
IF (NEWIND.LE.1) THEN
  LN2=LOG(2)
ENDIF
; Define parameters
KA=LN2/TABS
ALAG1=TLAG
S2=V

$DES
GUT=A(1)
DCP=A(2)/V
RATEIN=KA*GUT
DADT(1)= -RATEIN
DADT(2)= RATEIN - DCP*CL

$ERROR
CP=A(2)/V
IF (DVID.LE.1) THEN
  Y=CP*(1+RUVCV) + RUVSD
ENDIF


CE=CP
PCA=E0 + EMAX*CE/(C50+CE)
IF (DVID.EQ.2) THEN
  Y=PCA + RUVFX
ENDIF
```

©NHQ Holford, 2005, all rights reserved.

Note the variables defined in \$DES cannot have the same name as variables in other blocks e.g. you cannot define $CP=A(2)/V$ in \$DES and also in \$ERROR. In this example the variable name DCP is used in \$DES and the variable name CP is used in \$ERROR.

Slide
12

Warfarin Immediate




©NHQ Holford, 2005, all rights reserved.

Individual 'post hoc' predictions from the immediate effect model describe most of the time course of PCA but in general the time of the predicted greatest effect is earlier than the observed effect.

Slide
13

Warfarin Immediate Predictive Check



The visual predictive check shows clearly the systematic bias with the greatest effect coming too early.

Slide
14

Delayed Effect Effect Compartment Model

- Two Basic Approaches
 - \$PRED
 - Write model for CP and CE
 - PREDPP
 - ADVAN4 Very small peripheral compartment
 - ADVAN6 Differential Equations

The visual predictive check points to using a model with a delayed effect. The simplest form of delayed effect model uses an effect compartment. This is usually considered an empirical model but if the delay in effect is only a few minutes then it might be a reasonable way to describe the time course of drug distribution from plasma to the site of action in an organ.

\$PRED and ADVAN6 can be used to write exact solutions to the effect compartment model. Because ADVAN6 has to solve differential equations it is slower. ADVAN4 can be used by making the peripheral compartment have a negligible volume. This is helpful for complex dosing histories that would be hard to code in \$PRED.

Slide
15

Effect Compartment Using \$PRED

```

$PRED
; Get DOSE from AMT
IF (NEWIND.LE.1) THEN
  LN2=LOG(2)
  DOSE=0
ENDIF
IF (AMT.GT.0) DOSE=AMT

; Define parameters
KA=LN2/TABS
KE=CL/V
KEQ=LN2/TEQ

; Adjust time for lagtime
IF (TIME.LE.TLAG) THEN
  TNOW=0
ELSE
  TNOW=TIME-TLAG
ENDIF

; Plasma concentration
EXPKA=EXP(-KA*TNOW)
EXPKE=EXP(-KE*TNOW)
CP=DOSE*KA/(V*(KA-KE))*(EXPKE-EXPKA)
IF (DVID.LE.1) THEN
  Y=CP*(1+RUVFV) + RUVSD
ENDIF

; Effect compartment model concentration
EXPKEQ=EXP(-KEQ*TNOW)
CEEKKE=EXPKE/(KA-KE)/(KEQ-KE)
CEEKKA=EXPKA/(KE-KA)/(KEQ-KA)
CEEKKQ=EXPKEQ/(KA-KEQ)/(KE-KEQ)
CE=DOSE*KA*KEQ/V*(CEEKKE+CEEKKA+CEEKKQ)
PCA=E0 + EMAX*CE/(C50+CE)
IF (DVID.EQ.2) THEN
  Y=PCA + RUVFX
ENDIF

```

The effect compartment model for a first-order input one compartment model requires an additional exponential term.

Slide
16

Effect Compartment Using ADVAN4

```
$SUB ADVAN4 TRANS4
$PK
; Define LN2
IF (NEWIND.LE.1) THEN
  LN2=LOG(2)
ENDIF

; Define parameters
KA=LN2/TABS
ALAG1=TLAG
S2=V
V2=V

KEQ=LN2/TEQ
V3=V2*0.0001 ; negligible volume
Q=V3*KEQ

$ERROR
CP=F
IF (DVID.LE.1) THEN
  Y=CP*(1+RUVCV) + RUVSD
ENDIF

CE=A(3)/V3
PCA=E0 + EMAX*CE/(C50+CE)
IF (DVID.EQ.2) THEN
  Y=PCA + RUVFX
ENDIF
```

©1990 Habbert, 2005, all rights reserved.

Note how the peripheral compartment volume is scaled to be a very small fraction (0.0001) of the central compartment volume. This ensures it will have a negligible influence on the model predictions for the central compartment concentrations.

Slide
17

Effect Compartment Using ADVAN6

```
$SUBR ADVAN6 TOL=3
$MODEL
COMP (DEPOT)
COMP (CENTRAL)
COMP (EFFECT)

$PK
; Define LN2
IF (NEWIND.LE.1) THEN
  LN2=LOG(2)
ENDIF
; Define parameters
KA=LN2/TABS
ALAG1=TLAG
S2=V
KEQ=LN2/TEQ

$DES
GUT=A(1)
DCP=A(2)/V
DCE=A(3)

RATEIN=KA*GUT
DADT(1)= -RATEIN
DADT(2)= RATEIN - DCP*CL
DADT(3)= KEQ*(DCP - DCE)

$ERROR
CP=A(2)/V
IF (DVID.LE.1) THEN
  Y=CP*(1+RUVCV) + RUVSD
ENDIF


CE=A(3)
PCA=E0 + EMAX*CE/(C50+CE)
IF (DVID.EQ.2) THEN
  Y=PCA + RUVFX
ENDIF
```

©1990 Habbert, 2005, all rights reserved.

The solution to differential equation 3 is scaled in terms of concentration. There is no need to define a volume of this compartment. It has no meaning for this model.

Slide
18

Warfarin CE ADVAN4




©1990 Habbert, 2005, all rights reserved.

The individual 'post hoc' predictions are much better than using the immediate model. This model looks fine. But are we missing something?

Slide
19

Warfarin Ce ADVAN4 Predictive Check



The visual predictive check confirms that the average prediction matches the observed effect time course but the variability is clearly overestimated.

Slide
20

Turnover Using ADVAN6

```

$SUBR ADVAN6 TOL=3
$MODEL
  COMP (DEPOT)
  COMP (CENTRAL)
  COMP (TRNOVR)

$PK
  F3=E0
  KP3A=LN2/TEQ
  RP3A=E0*KP3A

$DES
  RATEIN=KA*A(1)
  DCP=A(2)/V
  DP3A=A(3)
  PD=1+EMAX*DCP/(C50+DCP)
  DADT(1)=-RATEIN
  DADT(2)=RATEIN - CL*DCP
  DADT(3)=RP3A*PD - KP3A*DP3A

$ERROR
  CP=A(2)/V
  IF (DVID.LE.1) THEN
    Y=CP*(1+RUVCV) + RUVSD
  ENDIF

  PCA=A(3)
  IF (DVID.EQ.2) THEN
    Y=PCA + RUVFX
  ENDIF

```

The turnover family of models describe delayed drug effects where the delay is due to the turnover of a physiological mediator. This is well understood as the mechanism of the delay for warfarin. The tricky part of this model is the need to modify the data set in order to initialize the PCA mediator compartment. This is done by using the F3 bioavailability fraction for the effect compartment.

Slide
21

Using Bioavailability Fraction to Initialize A Compartment

#ID	TIME	AMT	CMT
1	0	1	3
1	0	1.00	1

```

$MODEL
  COMP (DEPOT)
  COMP (CENTRAL)
  COMP (TRNOVR)


$PK
  F3=E0

```

Initial conditions can be defined by a combination of adding a special record to the data set and using the NM-TRAN bioavailability fraction. The special record is a dose record indicating a nominal AMT of 1 in the desired compartment. This record appears once for every individual at TIME=0. The bioavailability fraction for AMT values is specified using the Fn variable in \$PK (n is the number of the compartment e.g. F1 is the bioavailability for CMT=1). By setting the bioavailability fraction for the compartment, the amount in that compartment at time zero will be equal to AMT=1 multiplied by F and thus the desired initial condition at TIME=0 will be obtained. The example shows a turnover model with a physiological mediator being defined in compartment 3. At steady state (the initial condition), the concentration of mediator at TIME=0 is E0. A single record defining the AMT in the mediator compartment must be placed at the start of every individual record.

Slide
22

Warfarin Turnover




©NHQ Holford, 2005, all rights reserved.

Turnover model individual “post hoc” predictions

Slide
23

Effect Cpt vs Turnover




©NHQ Holford, 2005, all rights reserved.

The individual “post hoc” predictions from the turnover model do not look any better than the predictions from the effect compartment model.

Slide
24

Warfarin Turnover Predictive Check



©NHQ Holford, 2005, all rights reserved.

The visual predictive check shows that the turnover model describes the variability of the observations much better than the effect compartment model but still does not properly capture the terminal recovery phase in two subjects.

Slide
25

Comparison of Models

Run	Obj	Min	Covariance	Eval	Sig
ka1_to_emax_ADVAN6	1202.815	Terminated Inf Obj Func	NONE	657	.
ka1_to_emax1_ADVAN6	1202.816	Successful	OK	367	3.1
ka1_ce_emax_ADVAN4	1271.931	Successful	ABORTED	597	3.3
ka1_ce_emax_ADVAN6	1271.936	Successful	ABORTED	539	3.2
ka1_ce_emax_PRED	1271.939	Successful	OK	608	3.2
ka1_im_emax_ADVAN2	1442.809	Successful	ABORTED	1225	3.0
ka1_im_emax_ADVAN6	1442.809	Successful	OK	929	3.0
ka1_im_emax_PRED	1442.809	Successful	ABORTED	1384	3.5

©NNG Hofford, 2005, all rights reserved.

Fixing Emax to 1 allowed the turnover model to finish successfully.
Whether the covariance step runs is not a helpful criterion of model suitability.

Slide
26

Comparison of Parameters

Run	POP E0	POP EMAX	POP C50	POP TEQ
ka1_to_emax_ADVAN6	96.6	-1	1.18	13
ka1_to_emax1_ADVAN6	96.6	-1	1.18	13
ka1_ce_emax_ADVAN4	96.7	-256	11.1	39.5
ka1_ce_emax_ADVAN6	96.7	-256	11.1	39.5
ka1_ce_emax_PRED	96.7	-257	11.1	39.5
ka1_im_emax_ADVAN2	96.5	-74.3	0.22	.
ka1_im_emax_ADVAN6	96.5	-74.3	0.22	.
ka1_im_emax_PRED	96.5	-74.3	0.22	.

©NNG Hofford, 2005, all rights reserved.

The half-life of PCA is reported in the literature to be about 14 hours. This is very close to the TEQ estimate of 13 hours from this data set. Notice also the more physiological meaning of an Emax of -1 i.e. 100% inhibition of PCA formation.

Slide
27

Comparison of Random Effects


Run	PPV E0	PPV EMAX	PPV C50	PPV TEQ	RUV FX
ka1_to_emax_ADVAN6	0.0533	6.54E-07	0.445	0.104	3.82
ka1_to_emax1_ADVAN6	0.0533	0.00E+00	0.445	0.104	3.82
ka1_ce_emax_ADVAN4	0.0510	1.77E-05	0.224	0.269	3.81
ka1_ce_emax_ADVAN6	0.0510	2.93E-06	0.224	0.269	3.81
ka1_ce_emax_PRED	0.0510	1.78E-04	0.224	0.269	3.81
ka1_im_emax_ADVAN2	0.0000	1.73E-02	0.879	.	8.37
ka1_im_emax_ADVAN6	0.0000	1.72E-02	0.878	.	8.37
ka1_im_emax_PRED	0.0000	1.72E-02	0.878	.	8.37

©NNG Hofford, 2005, all rights reserved.

The residual error is very similar for the effect compartment and turnover models. The turnover model suggests that between subject differences in EC50 are more important than the value predicted using the effect compartment model.

Slide
28

Warfarin Predictive Check



The diagnostic benefits of the visual predictive check are seen clearly for the 3 PKPD models.

Slide
29

